```
(* FS 2013 , nb zur FourierTransformation,  W.Quapp *)

(* Vorspann:
 Auch LaplaceTransform gibt Auskunft ueber die Frequenz einer Schwingung *)

l1 = LaplaceTransform[Sin[a t], t, s]
l2 = LaplaceTransform[Cos[a t], t, s]
```

$$\frac{a}{a^2 + s^2}$$

$$\frac{s}{a^2 + s^2}$$

```
Plot[{{l1, l2} /. a → 3}, {s, 0, 9}]
```




```
(*  l1 startet bei 1/a, l2 hat bei a den maximalen Wert 1/2a *)


(* Vorspann2: DiracDelta *)


dela[t_, to_, a_] := 1 / (2 a) /; to - a ≤  t ≤ to + a
dela[t_, to_, a_] := 0       /;  t > to + a
dela[t_, to_, a_] := 0       /;  t < to - a
Plot[dela[t, 0, 0.25], {t, -1, 1}, PlotStyle → Thick, AxesLabel → {t, "del"}]
```



```
fuerPlot = Table[dela[t, 0, a], { a, 0.01, 0.05, 0.01}]
grau = Table[GrayLevel[i], {i, 0, 0.7, 0.7 / 4}]
Plot[Evaluate[fuerPlot], {t, -0.1, 0.1}, PlotStyle → grau,
 AxesLabel → {t, "del"}, PlotRange → All, AspectRatio → 1.5]
```

```
{dela[t, 0, 0.01], dela[t, 0, 0.02],
 dela[t, 0, 0.03], dela[t, 0, 0.04], dela[t, 0, 0.05]}
```

```
{GrayLevel[0.], GrayLevel[0.175],
 GrayLevel[0.35], GrayLevel[0.525], GrayLevel[0.7]}
```




```
(* Im Limes =: DiracDelta, ist aber keine normale Funktion *)
```

```
Integrate[Exp[2. t] 1 / (2 × 0.01), {t, 1. - 0.01, 1. + 0.01}]
Exp[2 t] /. t → 1.
```

7.38955

7.38906

```
LaplaceTransform[DiracDelta[t - to], t, s]
```

$e^{-s\,to}$ HeavisideTheta[to]

```
LaplaceTransform[DiracDelta[t - 1], t, s]
```

e^{-s}

In[2]:=
```
(* Delta ist keine interessant- 'malbare' Funktion *)
Plot[ DiracDelta[ω] , {ω, 0, Pi}]
```

Out[2]=




```
(* +++++++++++++ +++++++++++ +++++++++++++ ++++++++++++ +++++++++++  *)

(* Setze interessante Funktionen in FourierTransform ein *)

FourierTransform[Exp[-t^2 / 2], t, ω]

FourierTransform[Sin[a 2 Pi t], t, ω]

FourierTransform[Cos[a 2 Pi t], t, ω]

FourierTransform[1, t, ω]

FourierTransform[HeavisideTheta[t], t, ω]
```

$$e^{-\frac{\omega^2}{2}}$$

$$i \sqrt{\frac{\pi}{2}} \text{ DiracDelta}[-2 a \pi + \omega] - i \sqrt{\frac{\pi}{2}} \text{ DiracDelta}[2 a \pi + \omega]$$

$$\sqrt{\frac{\pi}{2}} \text{ DiracDelta}[-2 a \pi + \omega] + \sqrt{\frac{\pi}{2}} \text{ DiracDelta}[2 a \pi + \omega]$$

$$\sqrt{2 \pi} \text{ DiracDelta}[\omega]$$

$$\frac{i}{\sqrt{2 \pi}\ \omega} + \sqrt{\frac{\pi}{2}} \text{ DiracDelta}[\omega]$$

```
(* next Bsp. *)

√(2 π) *
  FourierTransform[UnitStep[t + ll] * UnitStep[-t + ll] / (2 ll), t, ω] // ExpToTrig
```

$$\frac{\text{Sin}[ll\ \omega]\ \text{UnitStep}[2\ ll]}{ll\ \omega}$$

In[3]:=
```
Plot[UnitStep[-t + 1] * UnitStep[t + 1], {t, -2, 2}, PlotStyle → Thick]
```

Out[3]=




```
Plot[{Sin[ll ω] UnitStep[2 ll] / (ll ω) /. ll → 2},
 {ω, -3 Pi, 3 Pi}, PlotRange → All, AxesLabel → {ω, f}]
```




```
(* Etwas aehnliches noch einmal mit diskreten Daten *)
```

```
data = Join[Table[0, {450}], Table[1, {100}], Table[0, {450}]];
ListPlot[data, Joined → True]
ListPlot[Abs[Fourier[data]], Joined → True, PlotRange → All]
```





```
usefull = Partition[Abs[Fourier[data]], 200];
ListPlot[usefull[[1]], Joined → True, PlotRange → All]
```



```
(* Ein RechteckImpuls ergibt unter der FourierTransformation eine
   stark gedaempfte Schwingung,
man beachte die durch ABS ins Positive gedrehten Werte *)

(*  next Cos[a t] abgeschnitten  *)
```

In[4]:= $\sqrt{2\pi}$ * FourierTransform[Cos[a t] UnitStep[t + ll] UnitStep[-t + ll], t, ω]

Out[4]= $$\frac{(2\, a\, \text{Cos}[ll\, \omega]\, \text{Sin}[a\, ll] - 2\, \omega\, \text{Cos}[a\, ll]\, \text{Sin}[ll\, \omega])\, \text{UnitStep}[2\, ll]}{a^2 - \omega^2}$$

In[5]:= TrigReduce[%] /. UnitStep[2 ll] → 1

Out[5]= $$\frac{a\, \text{Sin}[a\, ll - ll\, \omega] + \omega\, \text{Sin}[a\, ll - ll\, \omega] + a\, \text{Sin}[a\, ll + ll\, \omega] - \omega\, \text{Sin}[a\, ll + ll\, \omega]}{a^2 - \omega^2}$$

In[6]:= (* Also ergibt sich wieder eine gedaempfte Schwingung *)

Plot$\left[\left\{\frac{2\, (a\, \text{Cos}[ll\, \omega]\, \text{Sin}[a\, ll] - \omega\, \text{Cos}[a\, ll]\, \text{Sin}[ll\, \omega])}{a^2 - \omega^2}\right.\right.$ /. {ll → 2, a → 2}},

{ω, -3 Pi, 3 Pi}, PlotRange → All, AxesLabel → {ω, f}]

Out[6]=



Betrachte Welle in endlichem Zeitintervall fuer a = 20

```
data = Table[N[Sin[20 × 2 Pi n / 400]], {n, 400}];
ListPlot[data, Joined → True]
ListPlot[Abs[Fourier[data]], Joined → True, PlotRange → All]
```





Betrachte Addition zweier Wellen in endlichem Zeitintervall
fuer a_1 = 20 und a_2 = 80

```
data = Table[N[Sin[20 × 2 Pi n / 400] + Sin[80 × 2 Pi n / 400]], {n, 400}];
ListPlot[data, Joined → True]
ListPlot[Abs[Fourier[data]], Joined → True, PlotRange → All]
```






```
Betrachte Addition zweier Wellen in endlichem Zeitintervall
fuer  a_ 1 = 20 und a_ 2 = 80 mit 2 verschiedenen Intensitaeten
```

```
data = Table[N[1 * Sin[20 × 2 Pi n / 400] + 0.75 * Sin[80 × 2 Pi n / 400]], {n, 400}];
ListPlot[data, Joined → True]
ListPlot[Abs[Fourier[data]], Joined → True, PlotRange → All]
```






```
(* nb zu FourierReihen, vergleiche:  Mma-Buch, S.680

 next: stark verrauschte 2 Wellen mit verschiedenen Frequenzen,
und Intensitaeten: in der Zeitdastellung herrscht das Chaos
   zuerst glatt -- dann verrauscht *)
```

In[8]:=
```
data1 = Table[N[1 * Sin[30 × 2 Pi n / 250] + 0.6 * Sin[21 × 2 Pi n / 250]], {n, 250}];
b1 = ListPlot[data1[[Range[1, 100]]], Joined → True]
```

Out[9]=



In[15]:=
```
data = Table[N[
    1 * Sin[30 × 2 Pi n / 250] + 0.6 * Sin[21 × 2 Pi n / 250] + Random[ ] - 1 / 2], {n, 250}];
ListPlot[data, Joined → True]
ListPlot[Abs[Fourier[data]], Joined → True, PlotRange → All]
```

Out[16]=



Out[17]=



```
usefull = Partition[Abs[Fourier[data]], 80];
ListPlot[usefull[[1]], Joined → True, PlotRange → All]
```




```
(* Die verrauschten Daten koennen 'gefiltert' werden indem
    die inverse Fourier Transformation eingesetzt wird *)
```

In[18]:=
```
filter = Join[Table[1, {80}], Table[0, {170}]];
fdata = InverseFourier[ Fourier[data] filter ];
b2 = ListPlot[1.5 * Re[fdata[[Range[1, 100]]]], Joined → True, PlotRange → All]
```

Out[20]=



In[21]:=
```
Show[b1, b2]
```

Out[21]=



```
(* Der Vergleich zeigt, dass die Frequenzen sehr gut rekonstruiert sind *)
```

In[22]:=
```
filter2 = Join[Table[0, {14}], Table[N[Exp[-(x - 6)^2 / 6]], {x, 11}] ,
   Table[N[Exp[-(x - 6)^2 / 6]], {x, 11}] , Table[0, {214}]];
ListPlot[filter2[[Range[1, 60]]], Joined → True, PlotRange → All]

fdata = InverseFourier[ Fourier[data] filter2 ];
b3 = ListPlot[2 * Re[fdata[[Range[1, 100]]]], Joined → True, PlotRange → All]
```

Out[23]=



Out[25]=



In[26]:=
```
Show[b1, b3]
```

Out[26]=



Envelope of Beat Production

Beats are caused by the interference of two waves at the same point in space. The plot of the variation of resultant amplitude with time shows the periodic increase and decrease for two sine waves.

```
Plot[{Sin[10 t], Sin[50 t]}, {t, 0, 0.25 Pi}]
Plot[{2.02 Sign[Sin[10 t ]] Sqrt[Abs[Sin[10 t ]]],
  -0.275 * Sign[Sin[10 t ]] Sqrt[Abs[Sin[10 t ]]],
  Sin[10 t] + Sin[50 t]}, {t, 0, 0.25 Pi}]
(* in der letzten Zeile steht die Superposition der 2 Wellen *)
```





The image below is the beat pattern produced by a London police whistle, which uses two short pipes to produce a unique and piercing three - note sound.

```
g = Import["londbeat.gif"]
```